# Maintainer: Felix Yan <felixonmars@archlinux.org>

pkgname=python-pyte
pkgver=0.8.2
pkgrel=1
pkgdesc='Simple VTXXX-compatible terminal emulator'
arch=('any')
license=('LGPL3')
url='https://github.com/selectel/pyte'
depends=('python-wcwidth')
makedepends=('python-setuptools' 'python-pytest-runner')
source=("https://github.com/selectel/pyte/archive/$pkgver/$pkgname-$pkgver.tar.gz")
sha512sums=('e9e79babb81b32eedbb4f0585ef07f8978e8592171834d9ce43a09f8a1d3156c00abb535e887ba9c83909bf1010210009b9a00ccbec7c2e430e5147120949743')

build() {
  cd pyte-$pkgver
  python setup.py build
}

check() {
  cd pyte-$pkgver
  python setup.py pytest
}

package() {
  cd pyte-$pkgver
  python setup.py install --root="$pkgdir" --optimize=1
}

# vim:set ts=2 sw=2 et:
